LIBNAME CARDIA 'Q:\SAS\VER83\DATA';

* THE FOLLOWING IS BASED ON CODE FROM DR. DAVID JACOBS ON 01-18-2013 ;
* N=3393 ;

DATA CognitiveFuncScores (KEEP=ID HRAVLT_IMMED HRAVLT_SHORT HRAVLT_LONG HRAVLT_COMPMEM HDSST HSTROOP1 HSTROOP2 HSTROOP3 HSTROOPINT HSTROOPINTALT HSTROOPPROB HTOTSEC1 HTOTSEC2 HTOTSEC3 TOT1SEC TOT2SEC TOT3SEC TOT1CHECK TOT2CHECK TOT3CHECK);

MERGE CARDIA.H3F89 (KEEP=ID H89TST1PHYS H89TS1NCOP H89TS1VIS H89TS1HEAR H89TS1CNREAD H89TS2PHYS H89TS2NCOP H89TS2VIS H89TS2HEAR H89TS2CNREAD H89TS3PHYS H89TS3NCOP H89TS3VIS H89TS3HEAR H89TOTSEC1-H89TOTSEC3 H89COMPSEC1-H89COMPSEC3 H89COMPMIN1-H89COMPMIN3 H89NERR1-H89NERR3)
CARDIA.H3F88 (KEEP=ID H88TOTSC)
CARDIA.H3F87 (KEEP=ID H87SCORE1-H87SCORE7);

BY ID;

* THIS PARTICIPANT WITHDREW CONSENT TO ALL CARDIA DATA ;
IF ID=203232105195 THEN DELETE;

* DSST ;
HDSST=H88TOTSC;
IF H88TOTSC=0 THEN HDSST=.;

* RAVLT: RAVLT_LONG (LONG DELAY) IS RECOMMENDED FOR MOST PURPOSES ;
* IMMEDIATE IS THE MEASURE OF LEARNING, SHORT DELAY IS NOT MUCH OF A MEMORY TEST, COMPOSITE IS A SUMMARY OF LEARNING, SHORT MEMORY, AND LONG DELAY ;
HRAVLT_IMMED=MEAN(OF H87SCORE1-H87SCORE5);  * SHA: REGARDLESS OF HOW MANY SCORES ;
HRAVLT_SHORT=H87SCORE6;
HRAVLT_LONG=H87SCORE7;
* COMPOSITE MEMORY SCORE ;
HRAVLT_COMPMEM=(HRAVLT_IMMED+HRAVLT_SHORT+HRAVLT_LONG)/3; * SHA: IF ONE MISSING, ALL MISSING;

* STROOP;
* SET TO MISSING IF ANY PHYSICAL, COOPERATION, VISION, LITERACY PROBLEM (NO HEARING PROBLEMS OCCURRED);
* IT AFFECTS THE WHOLE SEQUENCE OF TESTS, N=37;
IF SUM(H89TST1PHYS, H89TST1NCOP, H89TST1VIS, H89TST1HEAR, H89TST1CNREAD, H89TST2PHYS, H89TST2NCOP, H89TST2VIS, H89TST2HEAR, H89TST2CNREAD, H89TST3PHYS, H89TST3NCOP, H89TST3VIS, H89TST3HEAR, H89TST3CNREAD)^= . THEN DO;
   HSTROOPPROB=1;
   H89TOTSEC1=.; H89TOTSEC2=.; H89TOTSEC3=.;
   H89COMPSEC1=.; H89COMPSEC2=.; H89COMPSEC3=.;
END;

* EDIT THE DIFFERENCE BETWEEN WHAT WAS WRITTEN DOWN FOR MINUTES AND SECONDS VS TOTAL SECONDS. USUALLY USE THE TOTAL SECONDS FIELD. WHEN THE SEQUENCE OF TEST TIMES DECREASES FROM TEST 1 TO TEST 2 OR FROM TEST 2 TO TEST 3, THERE MAY HAVE BEEN A RECORDING ERROR, SO IF THERE IS A DISAGREEMENT BETWEEN THE TWO RECORDING METHODS, USE THE MINUTE AND SECOND RECORDING METHOD. FOR ID 205438227085 THE TOTAL SECONDS WAS CORRECT.
THIS MADE A LARGE CORRECTION IN 5 PEOPLE;
TOT1SEC=60*H89COMPMIN1+H89COMPSEC1;
TOT2SEC=60*H89COMPMIN2+H89COMPSEC2;
TOT3SEC=60*H89COMPMIN3+H89COMPSEC3;
TOT1CHECK=TOT1SEC-H89TOTSEC1;
TOT2CHECK=TOT2SEC-H89TOTSEC2;
TOT3CHECK=TOT3SEC-H89TOTSEC3;
H89TOTSEC21=H89TOTSEC2-H89TOTSEC1;
H89TOTSEC32=H89TOTSEC3-H89TOTSEC2;
HTOTSEC1=H89TOTSEC1;
HTOTSEC2=H89TOTSEC2;
HTOTSEC3=H89TOTSEC3;
IF ID^=205438227085 THEN DO;
   IF .<H89TOTSEC21<0 AND TOT1CHECK^=0 THEN HTOTSEC1=TOT1SEC;
   IF .<H89TOTSEC21<0 AND TOT2CHECK^=0 THEN HTOTSEC2=TOT2SEC;
   IF .<H89TOTSEC32<0 AND TOT2CHECK^=0 THEN HTOTSEC2=TOT2SEC;
   IF .<H89TOTSEC32<0 AND TOT3CHECK^=0 THEN HTOTSEC3=TOT3SEC;
END;

* STROOP TEST SCORES CORRECTED FOR NUMBER OF ERRORS;
HSTROOP1=HTOTSEC1+H89NERR1;
HSTROOP2=HTOTSEC2+H89NERR2;
HSTROOP3=HTOTSEC3+H89NERR3;
* DELETE ANY TIME<=5 SECONDS;
IF .<HTOTSEC1<=5 THEN HSTROOP1=.;
IF .< HTOTSEC2<=5 THEN HSTROOP2=.;
IF .<HTOTSEC3<=5 THEN HSTROOP3=.;

* INTERFERENCE VARIABLE;
HSTROOPINT=HSTROOP3-HSTROOP2;

* ALTERNATE INTERFERENCE SCORE;
IF H89NERR3=40 OR H89NERR2=40 OR HTOTSEC3=. OR HTOTSEC2=. THEN HSTROOPINTALT=. ;
ELSE HSTROOPINTALT=(HTOTSEC3/(40-H89NERR3))-(HTOTSEC2/(40-H89NERR2));

* SET DSST AND RAVLT MISSING FOR IMPAIRED PEOPLE;
IF HSTROOPPROB=1 THEN DO;
    HRAVLT_IMMED=.;
    HRAVLT_SHORT=.;
    HRAVLT_LONG=.;
    HRAVLT_COMPMEM=.;
    HDSST=.;
END;

LABEL HRAVLT_IMMED = 'MEAN REY AUDITORY-VERBAL LEARNING TEST (RAVLT) - IMMEDIATE RECALL'
HRAVLT_SHORT = 'REY AUDITORY-VERBAL LEARNING TEST (RAVLT) - SHORT_DELAY FREE RECALL'
HRAVLT_LONG = 'REY AUDITORY-VERBAL LEARNING TEST (RAVLT) - LONG_DELAY FREE RECALL'
HRAVLT_COMPMEM = 'REY AUDITORY-VERBAL LEARNING TEST (RAVLT) - COMPOSITE MEMORY'
HDSST = 'DIGIT SYMBOL SUBSTITUTION TEST (DSST) SCORE - REVISED'
HSTROOP1 = 'SROOP TEST SCORE CORRECTED FOR NUMBER OF ERRORS - TEST1'
HSTROOP2 = 'SROOP TEST SCORE CORRECTED FOR NUMBER OF ERRORS - TEST2'
HSTROOP3 = 'SROOP TEST SCORE CORRECTED FOR NUMBER OF ERRORS - TEST3'
HSTROOPINT = 'SROOP TEST SCORE CORRECTED FOR NUMBER OF ERRORS - INTERFERENCE (HSTROOP3-HSTROOP2)'
HSTROOPINTALT = 'ALTERNATIVE SROOP TEST SCORE CORRECTED FOR NUMBER OF ERRORS - INTERFERENCE'
HSTROOPPROB = 'ANY PHYSICAL, COOPERATION, VISION, LITERACY PROBLEM (NO HEARING PROBLEMS OCCURRED)'
HTOTSEC1 = 'TIME FOR SROOP TEST COMPLETED (SEC) - TEST1 EDITED'
HTOTSEC2 = 'TIME FOR SROOP TEST COMPLETED (SEC) - TEST2 EDITED'
HTOTSEC3 = 'TIME FOR SROOP TEST COMPLETED (SEC) - TEST3 EDITED'
TOT1SEC = 'TIME FOR SROOP TEST COMPLETED (SEC) - TEST1 CONVERTED FROM H89COMPMIN1 AND H89COMPSEC1'
TOT2SEC = 'TIME FOR SROOP TEST COMPLETED (SEC) - TEST2 CONVERTED FROM H89COMPMIN2 AND H89COMPSEC2'
TOT3SEC = 'TIME FOR SROOP TEST COMPLETED (SEC) - TEST3 CONVERTED FROM H89COMPMIN3 AND H89COMPSEC3'
TOT1CHECK = 'TOT1SEC-H89TOTSEC1'
TOT2CHECK = 'TOT2SEC-H89TOTSEC2'
TOT3CHECK = 'TOT3SEC-H89TOTSEC3';
RUN;